

```

10 REM Income Tax and NI Calculator
20 REM (c) Computing with the Amstrad
30 REM ----- CPC -----
40 BORDER 0:INK 0,10:INK 1,0:INK 2,22:INK 3,6
50 WHILE 1
60 MODE 2
70 GOSUB 690 :REM Flush K/B
80 GOSUB 150 :REM Input details
90 GOSUB 230 :REM Calculate tax
100 GOSUB 350 :REM Calculate NI
110 GOSUB 430 :REM Chart
120 GOSUB 690
130 GOSUB 720 :REM Wait for key press
140 WEND
150 REM ----- Input details -----
160 WINDOW 20,80,10,25
170 INPUT "Please enter tax code (numbers only) : ",taxcode
180 INPUT "Taxable pay earned to date? : ",pay2date
190 INPUT "How much tax have you paid to date? : ",tax2date
200 INPUT "What's the tax week? : ",taxweek
210 INPUT "How much pay is due this week? : ",paydue
220 RETURN
230 REM ----- Calculate Tax -----
240 taxrate=0.27
250 paytot=pay2date+paydue
260 IF paytot>17900 THEN taxrate=0.4
270 IF paytot>20400 THEN taxrate=0.45
280 IF paytot>25400 THEN taxrate=0.5

```

```

290 IF paytot>33300 THEN taxrate=0.55
300 IF paytot>41200 THEN taxrate=0.6
310 taxfree=taxcode*10+5
320 taxdue=(paytot-((taxfree/52)*taxweek))*taxrate
330 tax2pay=(INT((taxdue-tax2date)*100))/100
340 RETURN
350 REM ----- Calculate NI -----
360 IF paydue<39 THEN nifrac=0
370 IF paydue>=39 AND paydue<65 THEN nifrac=0.05
380 IF paydue>=65 AND paydue<100 THEN nifrac=0.07
390 IF paydue>=100 AND paydue<295 THEN nifrac=0.09
400 IF paydue>=295 THEN nidue=25.65:RETURN
410 nidue=paydue*nifrac
420 RETURN
430 REM ----- Chart -----
440 MODE 1
450 ORIGIN 96,40
460 bw=138:vsf=350/paydue
470 bx=0:by=0:bd=350:bc=2:GOSUB 590:bc=1:GOSUB 640
480 bx=160:by=0:bd=tax2pay*vsf:bc=3:GOSUB 590:bc=1:GOSUB 640:bx=0:GOSUB 640
490 bx=320:by=0:bd=nidue*vsf:bc=3:GOSUB 590:bc=1:GOSUB 640:bx=0:by=tax2pay*vsf:GOSUB 640
500 LOCATE 7,24:PRINT "Gross Pay";TAB(17);"Tax";TAB(27);"NI"
510 LOCATE 7,25:a=paydue:GOSUB 750:PRINT

```

```

a$
520 LOCATE 17,25:a=tax2pay:GOSUB 750:PRINT a$
530 LOCATE 27,25:a=nidue:GOSUB 750:PRINT a$
540 LOCATE 27,2:PRINT "Total Tax+NI":LOCATE 27,3:PRINT "due this week"
550 LOCATE 27,4:a=tax2pay+nidue:GOSUB 750:PRINT a$
560 bx=300:by=284:bd=70:bw=240:GOSUB 640
570 MOVER 4,-4:DRAW 148,a*vsf:DRAW 296,290
580 RETURN
590 REM ----- Block Fill -----
600 FOR x=bx TO bx+bw STEP 2
610 MOVE x,by:DRAWR 0,bd,bc
620 NEXT
630 RETURN
640 REM ----- Box bx,by,bd -----
650 MOVE bx,by
660 DRAWR bw,0,bc:DRAWR 0,bd
670 DRAWR -bw,0:DRAWR 0,-bd
680 RETURN
690 REM ---- Flush Keyboard Buffer ----
700 WHILE INKEY$<>"":WEND
710 RETURN
720 REM ----- Wait For Keypress -----
730 WHILE INKEY$="" :WEND
740 RETURN
750 REM ----- Tidy Cash Display -----
760 a$=STR$(ROUND(a,2))
770 a$="£"+MID$(a$,2,LEN(a$)-1)
780 RETURN

```